


3. (35 puan) (PÇ-2,3,4,5)





Yandaki şekilde bir fabrikadaki su ihtiyacını depolayan ve karşılayan bir sistem görülmektedir. Sistemde bir adet sirkülasyon pompası (motoru), Tank-1 ve Tank2 olmak üzere 2 adet tank, suyun akışına elektronik olarak izin veren yada akışı kapatan 2 adet vana, Tank-2'deki su seviye ikazını veren 2 adet seviye sensörü mevcuttur.

Aşağıda çalışma prensibi anlatılan sistem için INT0, INT1, T0, T1 kesme yapılarını kullanarak 8051 komutlarıyla sistemi gerçekleyiniz

Cihaz	Pin	Açıklama
Sirkülasyon Pompası	P1.0	Lojik 1 çalışır Lojik 0 durur
Elektronik Vana-1	P1.1	Lojik 1 akış var Lojik 0 akış yok
Elektronik Vana-2	P1.2	Lojik 1 akış var Lojik 0 akış yok
İkaz Lambası	P1.3	Lojik 1 yanar, Lojik 0 söner
Seviye Sensörü-1	INT0 (P3.2)	Lojik 1 çizgi altında Lojik 0 çizgi üstünde
Seviye Sensörü-2	INT1 (P3.3)	Lojik 0 çizgi altında Lojik 1 çizgi üstünde

**Sistemin çalışma prensibi:**

- Motor başlangıçta durmaktadır. Harici kesme (INT1) girişine bağlı olan Seviye sensörü-2'den Lojik-0 sinyali gelir gelmez Vana-1 açılır ve Tank-1'e su dolmaya başlar.
- Vana-1'le birlikte zamanlayıcı-1 (T1) devreye girer ve 50 milisn sonra pompa çalışır ve su pompalar.
- Motorun çalışmasıyla birlikte Vana-2'de açılır ve motor Tank-2'ye su pompalar.
- Motorun çalışmasıyla birlikte ikaz lambası 250 mikrosn yanıp-sönerek motorun çalıştığını gösterir. Bunu sağlamak için T0'ı kullanınız.
- Tank-2'deki su seviyesi harici kesme (INT0) girişine bağlı Seviye sensörü-1'e ulaştığında sensörde Lojik-0 sinyali oluşur.
- Bu durumda hemen Vana-1 kapatılır ve zamanlayıcı (T1) devreye girer ve 25 milisn sonra motor durdurulur, Vana-2 kapatılır ve motor durduğu için de ikaz lambasının ikaz vermesi de durdurulur.
- Sistem tekrar devreye girmek üzere seviye sensörü-2'nin Lojik-0 sinyalini üretmesini bekler.



;Su Tanki Sorusu

org 0000h

sjmp basla

org 0003h ;INT0 kesmesi

ljmp INT0Kesmesi

org 000Bh ;T0 kesmesi

ljmp TMR0Kesmesi

org 0013h ;INT1 kesmesi

ljmp INT1Kesmesi

org 001Bh ;T1 kesmesi

ljmp TMR1Kesmesi

basla:

CLR P1.0 ;motor baslangicta duruyor

MOV IE, #8Fh ;kesme yetkilendirme

MOV TMOD, #12h ;timer0 mod2, timer1 mod1

MOV TH0, -250 ; 250 mikro sn

MOV TL0, -250

MOV TH1, HIGH(-50000)

MOV TL1, LOW(-50000) ;Timer-1 icin 50 milisn

CLR TF0

CLR TR0

CLR TF1

CLR TR1

SETB IT0

SETB IT1

CLR P1.3 ;ikaz lambasi sondur

SJMP \$

TMR0kesmesi:

CLR TF0

CPL P1.3 ;ikaz lambasini tersle

RETI

TMR1kesmesi: ; burasi hem motor calistirma hem de durdurma  
; sirasinda cilisiyor. bu sebeple calisma  
; kontrolu yapilmali

CLR TR1

CLR TF1

;motor duruyorsa

JB P1.0, MotorCalisiyor

SETB P1.0 ;pompa calismaya basladi

SETB P1.2 ; vana-2 acildi

CLR TF0

SETB TR0 ;TMR0 calismaya basladi flash yapiyor

SJMP atla

;motor calisiyorsa

MotorCalisiyor:

CLR P1.0 ;pompa durdu

CLR P1.2 ;vana-2 kapatildi

CLR TF0

CLR TR0 ;TMR0 durduruldu

atla:

RETI

INT0Kesmesi: ;tank doldu

CLR P1.1 ;vana-1 kapat

MOV TH1, HIGH(-25000)

MOV TL1, LOW(-25000) ;Timer-1 icin 25 milisn

SETB TR1

RETI

INT1Kesmesi: ;tank bosaldi su doldur

SETB P1.1 ;vana-1 açık tank-1 doluyor

MOV TH1, HIGH(-50000)

MOV TL1, LOW(-50000) ;Timer-1 icin 50 milisn

SETB TR1

RETI

END